


NASA/TM-97-206266

1N-02
1998
048742



Hypersonic Pitching-Moment Shift for Stardust Reentry Capsule Forebody

William A. Wood
Langley Research Center, Hampton, Virginia

October 1997

The NASA STI Program Office ... in Profile

Since its founding, NASA has been dedicated to the advancement of aeronautics and space science. The NASA Scientific and Technical Information (STI) Program Office plays a key part in helping NASA maintain this important role.

The NASA STI Program Office is operated by Langley Research Center, the lead center for NASA's scientific and technical information. The NASA STI Program Office provides access to the NASA STI Database, the largest collection of aeronautical and space science STI in the world. The Program Office is also NASA's institutional mechanism for disseminating the results of its research and development activities. These results are published by NASA in the NASA STI Report Series, which includes the following report types:

- **TECHNICAL PUBLICATION.** Reports of completed research or a major significant phase of research that present the results of NASA programs and include extensive data or theoretical analysis. Includes compilations of significant scientific and technical data and information deemed to be of continuing reference value. NASA counter-part of peer-reviewed formal professional papers, but having less stringent limitations on manuscript length and extent of graphic presentations.
- **TECHNICAL MEMORANDUM.** Scientific and technical findings that are preliminary or of specialized interest, e.g., quick release reports, working papers, and bibliographies that contain minimal annotation. Does not contain extensive analysis.
- **CONTRACTOR REPORT.** Scientific and technical findings by NASA-sponsored contractors and grantees.


- **CONFERENCE PUBLICATION.** Collected papers from scientific and technical conferences, symposia, seminars, or other meetings sponsored or co-sponsored by NASA.
- **SPECIAL PUBLICATION.** Scientific, technical, or historical information from NASA programs, projects, and missions, often concerned with subjects having substantial public interest.
- **TECHNICAL TRANSLATION.** English-language translations of foreign scientific and technical material pertinent to NASA's mission.

Specialized services that help round out the STI Program Office's diverse offerings include creating custom thesauri, building customized databases, organizing and publishing research results... even providing videos.

For more information about the NASA STI Program Office, see the following:

- Access the NASA STI Program Home Page at <http://www.sti.nasa.gov>
- E-mail your question via the Internet to help@sti.nasa.gov
- Fax your question to the NASA Access Help Desk at (301) 621-0134
- Phone the NASA Access Help Desk at (301) 621-0390
- Write to:
NASA Access Help Desk
NASA Center for AeroSpace Information
800 Elkridge Landing Road
Linthicum Heights, MD 21090-2934

NASA/TM-97-206266



Hypersonic Pitching-Moment Shift for Stardust Reentry Capsule Forebody

William A. Wood
Langley Research Center, Hampton, Virginia

National Aeronautics and
Space Administration

Langley Research Center
Hampton, Virginia 23681-2199

October 1997

Available from the following:

NASA Center for AeroSpace Information (CASI)
800 Elkridge Landing Road
Linthicum Heights, MD 21090-2934
(301) 621-0390

National Technical Information Service (NTIS)
5285 Port Royal Road
Springfield, VA 22161-2171
(703) 487-4650

Abstract

Aerodynamic coefficients are presented for perfect-gas and equilibrium-air solutions of the Navier-Stokes equations about the Stardust reentry-capsule forebody at Mach numbers of 4.6, 7, 8.5, and 10. A comparison with Newtonian-flow assumptions indicates a divergence of the aerodynamic coefficients from Newtonian-flow for Mach numbers less than 10. The static stability of the forebody is reduced by a factor of 2.5 with decreasing freestream Mach number between Mach 10 and 7.

Nomenclature

A	Area, m^2
C	Force or moment coefficient
C_P	Pressure coefficient
c	Reference length, m
e	Unit vector, m
h	Altitude, km
i, j	Computational indices
M	Mach number
n	Unit normal vector, m
P	Pressure, Pa
S	Reference area, m^2
T	Temperature, K
t	Time during trajectory, s
t	Tangent vector, m
V	Velocity, m/s
x, y, z	Cartesian body axes, m
α	Angle of attack, degrees
γ	Ratio of specific heats
ϕ, θ	Surface inclination angles, rad
ρ	Density, kg/m^3

Subscripts:

A	Axial force
E	Surface element
N	Normal force
M	Pitching moment
0_2	Post-shock stagnation value
∞	Freestream

Introduction

Stardust[1] is a comet sample-and-return mission slated for the end of the century. Trajectory calculations for Earth atmospheric reentry require a determination of the capsule aerodynamics. Of primary interest for the axisymmetric capsule are lift, drag, and pitching moment coefficients.

Laminar, thin-layer Navier-Stokes solutions are obtained at four trajectory points, Mach 4.6, 7, 8.5, and 10, for the Stardust forebody, assuming a base pressure equal to freestream static pressure. Use of this base pressure assumption at similar Mach-number conditions for the Commercial Experiment Transporter (COMET) reentry capsule has been shown to produce errors of one percent or less[2, 3]. A complete set of Stardust aerodynamics has been published by Mitcheltree *et al*[4], including a subset of the present data.

Navier-Stokes solutions are compared with results using Newtonian-flow assumptions, revealing a sudden shift from Newtonian-like flow at the highest altitude/Mach number point to a non-Newtonian surface pressure distribution at lower altitudes/Mach numbers. While the trend away from Newtonian-flow as the Mach number decreases is expected, of note for this case is how abrupt the departure is between Mach 7 and 10. Over this Mach number range, covering eight seconds of flight time, the forebody static stability decreases by a factor of 2.5.

Configuration

The physical configuration of the Stardust reentry-capsule forebody is a spherically-blunted 60 degree half-angle cone. The nose radius is 0.23 m, the shoulder radius is 0.02 m, and the base radius is 0.41 m.

The axisymmetric computational grid contains 30 streamwise cells and 64 cells normal to the body, in a structured quadrilateral framework. Grid adaption was performed to align with the bow shock and boundary layer.

For the three-dimensional cases, a singularity-free grid is employed with a 58 by 29 cell surface mesh and 48 cells normal to the body. Figure 1 displays a side view of the surface mesh. Only the starboard side of the vehicle was computed, with left-right symmetry assumed in the flowfield. Adaption was performed on this grid to align with the bow shock and boundary layer.

Computational Methods

Laminar, thin-layer Navier-Stokes calculations were performed using the Langley Aerothermodynamic Upwind Relaxation Algorithm (LAURA)[5, 6]. LAURA is a second-order accurate finite-volume approximate Riemann solver[7]. Previous applications to Earth reentry capsules include COMET[2, 3] and the Aeroassist Flight Experiment[8]. Both perfect-gas and equilibrium-air[9] models were employed.

Modified-Newtonian solutions based on impact theory for hypersonic vehicles were calculated using the code presented in the appendix. This code was written specifically to be compatible with LAURA.

Cases

Four freestream conditions are considered, Mach 4.6, 7, 8.5, and 10, at 5 degrees angle of attack. Complete freestream descriptions are contained in Table 1.




Figure 1: Side view of singularity-free surface mesh for Stardust reentry capsule forebody.

M_∞	h	T_∞	ρ_∞	V_∞	t
10.3	44.4	255	2.04×10^{-3}	3290	82
8.5	43.2	252	2.41×10^{-3}	2720	86
7.1	42.1	250	2.82×10^{-3}	2270	90
4.6	39.6	244	4.09×10^{-3}	1470	100

Table 1: Freestream conditions.

The Mach 4.6 solution is for a perfect gas, while equilibrium air is used for the Mach 7–10 results.

Three axisymmetric solutions were also obtained, at Mach 7 and 10, equilibrium air, and Mach 7, perfect gas.

All Newtonian results are for perfect gas.

Results

Axial-force coefficients are presented in Table 2 for all cases. The reference area is 0.5189 m^2 . For C_A , unmodified-Newtonian results with a base-pressure correction, as described in the appendix, are tabulated for comparison with the LAURA solutions. Excellent agreement is seen for axial force between the two methods for Mach 7–10. At Mach 4.6, 5 degrees angle of attack, the Newtonian axial force is 8.5 percent higher than the LAURA result, indicating a breakdown

method	α	M_∞				
		4.6	7	7 (pg)	8.5	10
LAURA	0	—	1.506	1.454	—	1.515
Newtonian	0	—	1.51	—	—	1.50
LAURA	5	1.41	1.477	—	1.496	1.498
Newtonian	5	1.53	1.50	—	1.50	1.49

Table 2: Axial-force coefficients.

of the assumptions behind impact theory at those conditions.

Looking at the effect of gas model on the Mach-7, axisymmetric solution, the perfect-gas C_A is 3.4 percent lower than the equilibrium-air result.

Normal-force coefficients calculated at 5 degrees angle of attack are plotted in Figure 2 versus freestream Mach number. Included with the LAURA results are modified-Newtonian solutions. The modified-Newtonian results are nearly independent of Mach number at $C_N = 0.04$. LAURA predicts the Newtonian result at Mach 10, but shows nearly a 50 percent decrease in normal force coefficient by Mach 7, with a similar number at Mach 4.6 of $C_N = 0.021$.

A similar shift in values occurs for the pitching-moment coefficient, Figure 3. At Mach 10 the LAURA solution agrees with modified-Newtonian, giving




Figure 2: Normal-force coefficient versus Mach number.




Figure 3: Pitching-moment coefficient versus Mach number.

$C_M = -0.014$. Between Mach 10 and 7 the static stability of the capsule is more than halved to -0.0056 . Again, the solution at Mach 4.6 has aerodynamics similar to Mach 7. This drop in pitching-moment coefficient between Mach 10 and 7 makes the reentry capsule 60 percent less stable in just 8 seconds of flight time. A reduction in static stability for this capsule would likely lead to increased amplitude of angle-of-attack oscillations. The reference length is 0.8128 m and the center of gravity is on the centerline 0.285 m back from the nosetip. Pitching moments are calculated about the center of gravity.

Reduction of Static Stability

Seeking the physical fluid phenomenon causing this abrupt decrease in stabilizing pitching-moment coefficient, symmetry-plane surface pressure coefficients are plotted in Figure 4 for the LAURA Mach 7 and 10 cases. The modified-Newtonian solution at Mach 10 is included as well, but the Mach 7 Newtonian solution is omitted because it agrees with the Mach 10 result to within 0.5 percent.

The trends in the LAURA Mach 10 solution are similar to the results of impact theory, with nearly flat profiles along the conical portion of the forebody. At Mach 7 the leeside shows more of a recompression than the Mach 10 results, though the profile is relatively flat. Leeside pressure coefficients are 5 percent higher at Mach 7 than Mach 10. The windside character at the two Mach




Figure 4: Symmetry-plane surface pressure coefficients at 5 degrees angle of attack.

numbers differs significantly, with the pressure dropping off significantly toward the shoulder at the lower Mach number. This change in character of the surface pressure distribution between Mach 10 and 7, an increase in the leeside pressure and a dropoff of the windside pressure toward the shoulder, is responsible for the reduction in pitching-moment coefficient at these hypersonic speeds.

Looking at surface pressure distributions, the Mach 7 and 10 LAURA surface-pressure coefficients are presented side-by-side in Figure 5. On the conical portion of the forebody the Mach 10 contours form predominantly radial lines, consistent with impact theory. At Mach 7, however, the surface C_P contours are highly curved, more consistent with a locally subsonic distribution. The patterns seen in Figure 5 suggest a change in the nature of the sonic bubble between Mach 7 and 10.

Figure 6 displays such a change in the nature of the sonic bubble. Plotted over a head-on view of the heatshield surface is the location of the sonic line, outside the boundary layer, for the Mach 7 and 10 LAURA solutions. At Mach 10 the sonic bubble contains the spherical nosepoint and the windward third of the forebody cone. By Mach 7, though, the sonic bubble encompasses two-thirds of the forebody, allowing significant three-dimensional relieving effects for the forebody flow, resulting in the surface pressure distributions previously compared in Figure 5.




Figure 5: Surface C_p contours for LAURA Mach 7 and 10 solutions, $\alpha = 5^\circ$.




Figure 6: Sonic bubble locations for LAURA Mach 7 and 10 solutions, $\alpha = 5^\circ$.

Summary of Results

Aerodynamic coefficients are calculated for the Stardust reentry-capsule forebody at Mach 4.6, 7, 8.5, and 10. Axial-force coefficients agree well with Newtonian-flow assumptions for Mach 7–10. Normal-force and pitching-moment coefficients agree with Newtonian at Mach 10, but reduce sharply in magnitude with decreasing Mach number, dropping more than half their values by Mach 7.

The reduction in static stability by 60 percent is investigated and found to correspond with a change in the inviscid sonic-bubble location. The sonic bubble grows with increasing Mach number to encompass two-thirds of the forebody by Mach 7, at 5 degrees angle of attack.

Appendix

A Modified-Newtonian Surface Pressure Calculator Compatible with LAURA[10]

Abstract

Modified-Newtonian surface pressure coefficients are calculated and integrated to obtain lift and drag coefficients. Three-dimensional geometries defined as a structured surface mesh are assumed, with freestream pressure imposed on shadowed portions of the geometry. Inputs required are the surface mesh, flow angle of attack, freestream Mach number, ratio of specific heats, and a normalizing length and area. Axial/normal and lift/drag force coefficients and the pitching moment coefficient are output. A plot file is created of the surface pressure coefficient. Compatibility with LAURA is emphasized and the source code is presented in a User's Manual format to aid in customization.

Governing Equations

Modified-Newtonian surface pressures (see Anderson[11] §3.2–3.5, or Bertin[12] §6.2–6.3), based on impact theory, are calculated according to,

$$\frac{C_P}{C_{P,max}} = \sin^2 \theta = \cos^2 \phi \quad (1)$$

where the pressure coefficient is defined as,

$$C_P = \frac{P - P_\infty}{\frac{1}{2}\rho_\infty V_\infty^2} \quad (2)$$

Straight-Newtonian is obtained in Eqn. 1 by setting $C_{P,max} = 2$. θ is the inclination angle of the surface to the freestream velocity vector, \mathbf{V}_∞ . ϕ is the angle between \mathbf{V}_∞ and the unit-normal to the surface, \mathbf{n} . ϕ and θ are related as $\theta = \frac{\pi}{2} - \phi$.

Eqn. 1 is evaluated at a point using the relation,

$$\cos \phi = -\mathbf{e}_V \cdot \mathbf{n} \quad (3)$$

where \mathbf{e}_V is the unit vector along \mathbf{V}_∞ , so that $\mathbf{V}_\infty = V_\infty \mathbf{e}_V$.

Initializations

The surface is read as a structured mesh in i and j . x , y , and z are the nodal coordinates.




Figure 7: Orientation of angles for Newtonian flow solver.

```

program newtonian
parameter ( imx = 59, jmx = 30, kmx = 1 )
parameter ( itdim = 2 * (imx - 1) * (jmx - 1) )
dimension x(imx,jmx,kmx), y(imx,jmx,kmx), z(imx,jmx,kmx)
dimension dxi(imx,jmx), dyi(imx,jmx), dzi(imx,jmx)
dimension dxj(imx,jmx), dyj(imx,jmx), dzj(imx,jmx)
dimension ev(3), sn(imx,jmx,3)
dimension cpcpmax(imx,jmx), cp(imx,jmx)
dimension tarea(itdim), tnormal(itdim,3), tcpcpm(itdim)
dimension tcp(itdim), txyz(itdim,3)
real mag
integer diagnos
diagnos = 0

```

From the angle-of-attack, α , obtain the unit vector, e_V , defining the freestream velocity direction. In LAURA coordinates.

```

write (6,*) 'Enter angle-of-attack in degrees'
read (5,*) alpha
pi = acos( -1. )
alpha = alpha * pi / 180.
ev(3) = -cos( alpha )
ev(1) = sin( alpha )
ev(2) = 0.

```

Read the surface grid in single-block PLOT3D ASCII 3-D whole format.

```

open ( 10, file = 'newton.g', form = 'formatted' )
read (10,*) idim, jdim, kdim
if ( idim .gt. imx .or. jdim .gt. jmx .or. kdim .gt. kmx ) then
write (6,*) 'Increase dimensions: ', idim, jdim, kdim

```

```

      stop
    end if
    read (10,*) ((( x(i,j,k), i=1, idim), j=1, jdim), k=1, kdim),
$          ((( y(i,j,k), i=1, idim), j=1, jdim), k=1, kdim),
$          ((( z(i,j,k), i=1, idim), j=1, jdim), k=1, kdim)

```

Surface Normals at a Point

\mathbf{n} is computed at each surface grid node by taking the cross product of the surface tangents in i and j .

$$\mathbf{n} = \frac{\mathbf{t}_i \times \mathbf{t}_j}{\|\mathbf{t}_i \times \mathbf{t}_j\|} \quad (4)$$

Surface tangents are computed with second-order discrete derivative approximations, with no correction for grid stretching.

$$\mathbf{t}_i = (\Delta x_i, \Delta y_i, \Delta z_i)$$

At interior nodes,

$$\Delta x_i = \frac{x_{i+1} - x_{i-1}}{2}$$

At i_{min} or i_{max} ,

$$\Delta x_1 = \frac{-3x_1 + 4x_2 - x_3}{2}$$

Do i -direction derivatives from min to max.

```

do j = 1, jdim
  dxi(1,j) = ( -3. * x(1,j,1) + 4. * x(2,j,1) - x(3,j,1) )/2.
  dyi(1,j) = ( -3. * y(1,j,1) + 4. * y(2,j,1) - y(3,j,1) )/2.
  dzi(1,j) = ( -3. * z(1,j,1) + 4. * z(2,j,1) - z(3,j,1) )/2.
end do
do i = 2, idim - 1
  do j = 1, jdim
    dxi(i,j) = ( x(i+1,j,1) - x(i-1,j,1) )/2.
    dyi(i,j) = ( y(i+1,j,1) - y(i-1,j,1) )/2.
    dzi(i,j) = ( z(i+1,j,1) - z(i-1,j,1) )/2.
  end do
end do
do j = 1, jdim
  dxi(idim,j) = ( 3. * x(idim,j,1) - 4. * x(idim-1,j,1)
$             + x(idim-2,j,1) )/2.
  dyi(idim,j) = ( 3. * y(idim,j,1) - 4. * y(idim-1,j,1)
$             + y(idim-2,j,1) )/2.
  dzi(idim,j) = ( 3. * z(idim,j,1) - 4. * z(idim-1,j,1)
$             + z(idim-2,j,1) )/2.
end do

```

Do j -direction derivatives. If $i = 1$ is a singularity point, set $i_{min} = 2$ and assume $\mathbf{t}_j|_{i=1} = \mathbf{t}_j|_{i=2}$.

```

imin = 1
do i = imin, idim
  dxj(i,1) = ( -3. * x(i,1,1) + 4. * x(i,2,1) - x(i,3,1) )/2.
  dyj(i,1) = ( -3. * y(i,1,1) + 4. * y(i,2,1) - y(i,3,1) )/2.
  dzj(i,1) = ( -3. * z(i,1,1) + 4. * z(i,2,1) - z(i,3,1) )/2.
end do
do j = 2, jdim - 1
  do i = imin, idim
    dxj(i,j) = ( x(i,j+1,1) - x(i,j-1,1) )/2.
    dyj(i,j) = ( y(i,j+1,1) - y(i,j-1,1) )/2.
    dzj(i,j) = ( z(i,j+1,1) - z(i,j-1,1) )/2.
  end do
end do
do i = imin, idim
  dxj(i,jdim) = ( 3. * x(i,jdim,1) - 4. * x(i,jdim-1,1)
$   + x(i,jdim-2,1) )/2.
  dyj(i,jdim) = ( 3. * y(i,jdim,1) - 4. * y(i,jdim-1,1)
$   + y(i,jdim-2,1) )/2.
  dzj(i,jdim) = ( 3. * z(i,jdim,1) - 4. * z(i,jdim-1,1)
$   + z(i,jdim-2,1) )/2.
end do
if ( imin .eq. 2 ) then
  write (6,*) 'handling i=1 as pole'
  do j = 1, jdim
    dxj(1,j) = dxj(2,j)
    dyj(1,j) = dyj(2,j)
    dzj(1,j) = dzj(2,j)
  end do
end if

```

Form the surface normals **n**.

```

do i = 1, idim
  do j = 1, jdim
    sn(i,j,1) = dyi(i,j) * dzj(i,j) - dzi(i,j) * dyj(i,j)
    sn(i,j,2) = dzi(i,j) * dxj(i,j) - dxi(i,j) * dzj(i,j)
    sn(i,j,3) = dxi(i,j) * dyj(i,j) - dyi(i,j) * dxj(i,j)
    mag = sqrt(sn(i,j,1)**2 + sn(i,j,2)**2 + sn(i,j,3)**2)
    sn(i,j,1) = sn(i,j,1) / mag
    sn(i,j,2) = sn(i,j,2) / mag
    sn(i,j,3) = sn(i,j,3) / mag
  end do
end do

```


$C_P/C_{P,max}$ at a Point

If the point is shadowed, $\phi > \frac{\pi}{2}$, then $C_P = 0$. Otherwise apply Eqns. 1 and 3.

```

do i = 1, idim
  do j = 1, jdim
    cosphi = - ( ev(1) * sn(i,j,1) + ev(2) * sn(i,j,2) +
$      ev(3) * sn(i,j,3) )
    cosphi = max( 0., cosphi )
    cpcpmax(i,j) = cosphi**2
  end do
end do

```

$C_{P,max}$

Compute $C_{P,max}$ assuming the surface pressure equals the total pressure behind a normal shock in Eqn. 2. Using perfect gas assumptions,

$$C_{P,max} = \frac{P_{02} - P_{\infty}}{\frac{1}{2}\rho_{\infty}V_{\infty}^2} = \frac{2P_{\infty}}{\rho_{\infty}V_{\infty}^2} \left(\frac{P_{02}}{P_{\infty}} - 1 \right) = \frac{2}{\gamma M_{\infty}^2} \left(\frac{P_{02}}{P_{\infty}} - 1 \right)$$

$$\frac{P_{02}}{P_{\infty}} = \frac{(\gamma + 1)^{\frac{\gamma+1}{\gamma-1}} M_{\infty}^2}{2 \left[2 \left(2\gamma - \frac{\gamma-1}{M_{\infty}^2} \right) \right]^{\frac{1}{\gamma-1}}}$$

```

write (6,*) 'Enter Mach number and gamma'
read (5,*) amach, gamma
amach = amach**2
g1 = gamma - 1
g2 = gamma + 1
g3 = gamma
p02pi = g2**(g2/g1) * amach * 0.5 /
$   ( 2. * ( 2. * g3 - g1 / amach ) )** (1./g1)
cpmax = 2. / g3 / amach * ( p02pi - 1. )
if ( cpmax .lt. 1. .or. cpmax .gt. 2. ) then
  write (6,*) '*** Problem with Cpmax = ', cpmax
else
  write (6,*) 'Cpmax = ', cpmax
end if
do i = 1, idim
  do j = 1, jdim
    cp(i,j) = cpcpmax(i,j) * cpmax
  end do
end do

```

Write TECPLOT[13] data file.

```

open ( 11, file = 'newton.dat', form = 'formatted' )
write (11,*) 'TITLE = "Modified Newtonian Surface Pressure',
$      ' ratios"'
write (11,*) 'VARIABLES = "X", "Y", "Z", "Cp/Cpmax", "Cp"'
write (11,*) 'ZONE T = "Surface", I=', idim, ' J=', jdim,
$      ' F=POINT'
do j = 1, jdim
do i = 1, idim
write (11,*) x(i,j,1), y(i,j,1), z(i,j,1), cpcpmax(i,j),
$      cp(i,j)
end do
end do

```

Integration of Surface Pressures

Integrate the pressure distribution over the surface to obtain forces and moments. To do so, C_P is needed to be defined over surface area elements, not just at surface points.

Surface Areas and Normals

Triangulate each quadrilateral surface cell $[(i, j), (i+1, j), (i+1, j+1), (i, j+1)]$ using a diagonal cut to create triangular cells $[(i, j), (i+1, j), (i, j+1)]$ and $[(i+1, j+1), (i, j+1), (i+1, j)]$. Surface normals are computed using Eqn. 4. Surface tangents are computed using first-order differences.

```

icell = 0
do i = imin, idim - 1
do j = 1, jdim - 1
icell = icell + 1
txyz(icell,1) = ( x(i+1,j,1) + x(i,j,1) + x(i,j+1,1) ) / 3.
txyz(icell,2) = ( y(i+1,j,1) + y(i,j,1) + y(i,j+1,1) ) / 3.
txyz(icell,3) = ( z(i+1,j,1) + z(i,j,1) + z(i,j+1,1) ) / 3.
tdxi = x(i+1,j,1) - x(i,j,1)
tdyi = y(i+1,j,1) - y(i,j,1)
tdzi = z(i+1,j,1) - z(i,j,1)
tdxj = x(i,j+1,1) - x(i,j,1)
tdyj = y(i,j+1,1) - y(i,j,1)
tdzj = z(i,j+1,1) - z(i,j,1)
tnormal(icell,1) = tdyi * tdzj - tdzi * tdyj
tnormal(icell,2) = tdzi * tdxj - tdxj * tdzj
tnormal(icell,3) = tdxj * tdyj - tdyj * tdxj
tarea(icell) = 0.5 * sqrt( tnormal(icell,1)**2 +
$      tnormal(icell,2)**2 + tnormal(icell,3)**2 )
if ( tarea(icell) .lt. 0. ) then
write (6,*) 'problem: tarea(', icell, ') = ', tarea(icell)

```

```

        stop
    end if
    tnormal(icell,1) = tnormal(icell,1) / 2. / tarea(icell)
    tnormal(icell,2) = tnormal(icell,2) / 2. / tarea(icell)
    tnormal(icell,3) = tnormal(icell,3) / 2. / tarea(icell)
end do
end do
do i = idim - 1, 1, -1
    do j = jdim - 1, 1, -1
        icell = icell + 1
        txyz(icell,1) = ( x(i,j+1,1) + x(i+1,j+1,1) + x(i+1,j,1))/3.
        txyz(icell,2) = ( y(i,j+1,1) + y(i+1,j+1,1) + y(i+1,j,1))/3.
        txyz(icell,3) = ( z(i,j+1,1) + z(i+1,j+1,1) + z(i+1,j,1))/3.
        tdx1 = x(i,j+1,1) - x(i+1,j+1,1)
        tdy1 = y(i,j+1,1) - y(i+1,j+1,1)
        tdz1 = z(i,j+1,1) - z(i+1,j+1,1)
        tdxj = x(i+1,j,1) - x(i+1,j+1,1)
        tdyj = y(i+1,j,1) - y(i+1,j+1,1)
        tdzj = z(i+1,j,1) - z(i+1,j+1,1)
        tnormal(icell,1) = tdy1 * tdzj - tdz1 * tdyj
        tnormal(icell,2) = tdz1 * tdxj - tdx1 * tdzj
        tnormal(icell,3) = tdx1 * tdyj - tdy1 * tdxj
        tarea(icell) = 0.5 * sqrt( tnormal(icell,1)**2 +
$           tnormal(icell,2)**2 + tnormal(icell,3)**2 )
        if ( tarea(icell) .lt. 0. ) then
            write (6,*) 'problem: tarea(',icell,') = ', tarea(icell)
            stop
        end if
        tnormal(icell,1) = tnormal(icell,1) / 2. / tarea(icell)
        tnormal(icell,2) = tnormal(icell,2) / 2. / tarea(icell)
        tnormal(icell,3) = tnormal(icell,3) / 2. / tarea(icell)
    end do
end do

```

$C_P/C_{P,max}$ over an Area

Compute surface pressure coefficient in a manner analogous to the computation at a grid point.

```

icellmax = 2 * (idim - 1) * (jdim - 1)
if ( icellmax .ne. icell ) then
    write (6,*) 'problem, icellmax=',icellmax,'icell=',icell
    stop
end if
if ( icellmax .gt. itdim ) then
    write (6,*) 'problem, icellmax=',icellmax,'itdim=',itdim

```

```

    stop
end if
do ic = 1, icellmax
    cosphi = - ( ev(1) * tnormal(ic,1) + ev(2) * tnormal(ic,2)
$      + ev(3) * tnormal(ic,3) )
    cosphi = max( 0., cosphi )
    tcpcpm(ic) = cosphi**2
    tcp(ic) = tcpcpm(ic) * cpmax
    if (diagnos .eq. 2) write (6,*) 'Cell=', ic, ' Cp=', tcp(ic)
end do

```

Force Coefficients

The force coefficients are defined for a resultant force F as,

$$C_F = \frac{F}{\frac{1}{2}\rho_\infty V_\infty^2 S}$$

where S is the normalizing area. Assuming P_∞ for the base pressure, the force coefficient can be found as,

$$C_F = \sum_{\text{Elements } E} \frac{-C_P A_E \mathbf{n}_E \cdot \mathbf{i}_F}{S}$$

where A_E is the area of the surface element and \mathbf{i}_F is the unit vector in the direction of F . Using LAURA coordinates.

```

write (6,*) 'Enter area S to normalize force coefficients'
read (5,*) sarea
ca = 0.
cn = 0.
do ic = 1, icellmax
    ca = ca + tcp(ic) * tarea(ic) * tnormal(ic,3)    ! Axial
    cn = cn - tcp(ic) * tarea(ic) * tnormal(ic,1)    ! Normal
end do
ca = ca / sarea
cn = cn / sarea
cd = ca * cos( alpha ) + cn * sin( alpha )          ! Drag
cl = cn * cos( alpha ) - ca * sin( alpha )          ! Lift
write (6,*) 'Force Coefficients:'
write (6,*) 'C_A = ', ca
write (6,*) 'C_N = ', cn
write (6,*) 'C_D = ', cd
write (6,*) 'C_L = ', cl

```

Base Pressure Correction

Base pressures are corrected for the axial force coefficient using the method in APAS of Bonner *et al*[14].

$$\Delta C_A = \frac{1}{M_\infty^2} - \frac{0.57}{M_\infty^4}$$

```

cacor = 1. / amach - 0.57 / amach / amach
ca = ca + cacor
cd = ca * cos( alpha ) + cn * sin( alpha )      ! Drag
cl = cn * cos( alpha ) - ca * sin( alpha )      ! Lift
write (6,*) 'Force Coefficients with base pressure correction:'
write (6,*) 'C_A = ', ca
write (6,*) 'C_N = ', cn
write (6,*) 'C_D = ', cd
write (6,*) 'C_L = ', cl

```

Pitching Moment

The pitching-moment coefficient is defined as,

$$C_M = \frac{M}{\frac{1}{2}\rho_\infty V_\infty^2 S c}$$

where c is a normalizing length. The moment coefficient is computed about the center of gravity (x_{a_0}, x_{n_0}) as,

$$C_M = \frac{1}{S c} \sum_E -C_P A_E [(\mathbf{n}_E \cdot \mathbf{i}_a)(x_{n_E} - x_{n_0}) - (\mathbf{n}_E \cdot \mathbf{i}_n)(x_{a_E} - x_{a_0})]$$

In LAURA coordinates, the moment formula becomes,

$$C_M = \frac{1}{S c} \sum_E C_P A_E [(\mathbf{n}_E \cdot \mathbf{i}_z)(x_E - x_0) - (\mathbf{n}_E \cdot \mathbf{i}_x)(z_E - z_0)]$$

```

write (6,*) 'Enter cg location in LAURA coordinates, (x,z)'
read (5,*) x0, z0
write (6,*) 'Enter normalizing length for moment'
read (5,*) alength
cm = 0.
do ic = 1, icellmax
  cm = cm + tcp(ic) * tarea(ic) * (
$      tnormal(ic,3) * ( txyz(ic,1) - x0 ) -
$      tnormal(ic,1) * ( txyz(ic,3) - z0 ) )
end do
cm = cm / sarea / alength
write (6,*) 'Moment coefficient, nose up positive:', cm

stop
end

```

Acknowledgments

This document was prepared with L^AT_EX_{2 ϵ} [15] using EMACS[16] software. The Newtonian flow code presented in the appendix is a self-documented FORTRAN code processed through the F2LATEX program created by William L. Kleb.

References

- [1] Atkins, K. L., Brownlee, D. E., Duxbury, T., Yen, C., and Tsou, P., "STAR-DUST: Discovery's Interstellar Dust and Cometary Sample Return Mission," *Proceedings from the 1997 IEEE Aerospace Conference*, Feb. 1997.
- [2] Wood, W. A., Gnoffo, P. A., and Rault, D. F. G., "Aerodynamic Analysis of Commercial Experiment Transporter Re-Entry Capsule," *Journal of Spacecraft and Rockets*, Vol. 33, No. 5, Sep. 1996, pp. 643-646.
- [3] Wood, W. A., Gnoffo, P. A., and Rault, D. F. G., "Aerothermodynamic Analysis of Commercial Experiment Transporter (COMET) Reentry Capsule," AIAA Paper 96-0316, Jan. 1996.
- [4] Mitcheltree, R., Wilmoth, R., Cheatwood, F., Rault, D., Brauckmann, G., and Greene, F., "Aerodynamics of Stardust Sample Return Capsule," AIAA Paper 2304, Jun. 1997.
- [5] Gnoffo, P. A., Gupta, R. N., and Shinn, J. L., "Conservation Equations and Physical Models for Hypersonic Air Flows in Thermal and Chemical Nonequilibrium," NASA TP 2867, Feb. 1989.
- [6] Gnoffo, P. A., "An Upwind-Biased, Point-Implicit Relaxation Algorithm for Viscous, Compressible Perfect-Gas Flows," NASA TP 2953, February 1990.
- [7] Roe, P. L., "Approximate Riemann Solvers, Parameter Vectors, and Difference Schemes," *Journal of Computational Physics*, Vol. 43, October 1981, pp. 357-372.
- [8] Gnoffo, P. A., "Code Calibration Program in Support of the Aeroassist Flight Experiment," *Journal of Spacecraft and Rockets*, Vol. 27, No. 2, March-April 1990, pp. 131-142.
- [9] Liu, Y. and Vinokur, M., "Equilibrium Gas Flow Computations. I. Accurate and Efficient Calculation of Equilibrium Gas Properties," AIAA Paper 89-1736, Jun. 1989.
- [10] Cheatwood, F. M. and Gnoffo, P. A., "User's Manual for the Langley Aerothermodynamic Upwind Relaxation Algorithm (LAURA)," NASA TM 4674, Apr. 1996.

- [11] Anderson, J. D., *Hypersonic and High Temperature Gas Dynamics*, McGraw-Hill, Inc., 1989.
- [12] Bertin, J. J., *Hypersonic Aerothermodynamics*, Education Series, AIAA, Washington, DC, USA, 1994.
- [13] Amtec Engineering, Inc., Bellevue, Washington, *Tecplot User's Manual: Version 7*, Aug. 1996.
- [14] Bonner, E., Clever, W., and Dunn, K., "Aerodynamic Preliminary Analysis System II. Part I—Theory," NASA CR 182076, Apr. 1991.
- [15] Lamport, L., *TEX: A Document Preparation System—User's Guide and Reference Manual*, Addison-Wesley Publishing Company, New York, 1994.
- [16] Stallman, R., *GNU Emacs Manual*, Free Software Foundation, Inc., Cambridge, MA, tenth ed., Jul. 1994.

